Macaulay2で不変式環の計算(2)

前回のページ

今回はMacaulay2を使ってDu Val特異点に関する計算をしよう！

不変式環の計算

$SL_2(\mathbb{C})$の有限部分群による商多様体$\mathbb{C}^2/G$に現れる特異点をDu Val特異点と呼ぶ。$SL_2(\mathbb{C})$の有限部分群は分類されており、具体的にどの行列で生成されるかも分かっている。

参考文献：Graham Leuschke「The McKay correspondence」のp.15

Du Val特異点の座標環を計算してみよう。例としてD_4型を考える。この場合、対応する有限群は以下の二つの行列で生成される。

$$\begin{pmatrix} i & 0 \\ 0 & -i \end{pmatrix}, \quad \begin{pmatrix} 0 & i \\ i & 0 \end{pmatrix}$$

M2で計算するために、以下のように準備をする。

```
i1 : loadPackage "InvariantRing"  -- InvariantRingパッケージを読み込む

o1 = InvariantRing

o1 : Package

i2 : L = toField(QQ[w]/(w^2+1))  -- QQに1の4乗根を添加

o2 = L

o2 : PolynomialRing
```

```
i3 : R = L[x,y]

o3 = R

o3 : PolynomialRing
```

次に、群作用を定義し不変式環を計算する。

```
i21 : X1 = matrix{{w,0},{0,-w}}; X2 = matrix{{0,w},{w,0}};

                       2              2
o21 : Matrix (L[u..w])  <-- (L[u..w])

                       2              2
o22 : Matrix (L[u..w])  <-- (L[u..w])

i23 : action1 = finiteAction({X1,X2},R)  -- 二つの行列で生成され

o23 = R <- {| w 0  |, | 0 w |}
            | 0 -w |  | w 0 |

o23 : FiniteGroupAction

i24 : A = invariantRing action1

o24 =    4    4   2 2   5       5
      L[x  + y , x y , x y - x*y ] -- バージョンに依ってはこのよう

o24 : RingOfInvariants

i25 : definingIdeal A  -- 不変式環を多項式環のイデアルによる商として

             2       3    2
o25 = ideal(u u  - 4u  - u )
             1 2     2    3
```

```
o25 : Ideal of L[u ..u ]
                  1   3
```

最後の計算結果により、このケースの商多様体は、方程式$u_1^2u_2 - 4u_2^3 - u_3^2 = 0$で定義される$\mathbb{C}^3$の中のアフィン代数多様体と同型になることが分かった。座標変換

$$x = u_3, \quad z = 4^{1/3}u_2, \quad y = (-1)^{1/2}4^{1/6}u_1$$

を施すことで、よく知られたD_4型特異点の式$x^2 + y^2z + z^3 = 0$になる。

演習

1. D_6型の不変式環をM2で求め、商多様体が$x^2 + y^2z + z^5 = 0$で定義されるアフィン代数多様体と同型となることを確認せよ。
2. E_6型、E_7型でも同様に確認せよ。（少し計算に時間がかかるかもしれない。）

爆発の計算

まず、以下のコードをコピペしてM2に貼り付けてEnterキーを押そう。

```
affineCharts = S ->(
      -- affine charts of a blowup without simplification
      T := (flattenRing S)_0;
      U := ambient T;
      I := ideal T;
      varsOfS := apply(flatten entries vars S,i->sub(i,U));
      apply(varsOfS, i->  U / (I + ideal(i - 1)))) ;

isBlowupSmooth = S ->(
    -- checks if the Rees algebra of an ideal is smooth.
    all(affineCharts(S),isSmooth2)  ) ;

isSmooth2 = R -> (
  -- checks if an affine ring is smooth
  if (isPolynomialRing R) then true else
    (
      S := ambient R;
```

```
        I := ideal R;
        (ideal singularLocus I) == ideal(1_S)
      )
    ) ;

isBlowupNormal = S ->(
   -- checs if the Rees algebra of an ideal is normal.
      all(affineCharts(S),isNormal)) ;
```

これらは、安田が以前に自分の研究用に書いたMacaulay2関数の一部で、こちらで公開している。（ただし、公開している関数「affineCharts」は動かなくなっていたので、「affineCharts2」を今回「affineCharts」と置き換えた。「isSmooth」は同名の組み込み関数とバッティングするので、「isSmooth2」に変更した。）

例として、関数「isSmooth」が何をしているか説明する。入力として`R`を受け取ると（`R`は多項式環のイデアルによる商環を想定している）、まず`R`が多項式環かどうか判定し、多項式環なら`true`を返す。`R`が多項式環でない場合は、`R`を定義するのに使った多項式環を`S`、イデアルを`I`と置き、`ideal singuarLocus I`でsingular locusの定義イデアルを求め、それが自明なイデアル`ideal(1_S)`と一致するかどうかを確認する。M2での関数の作り方の詳細は、例えば「Macaulay2による計算代数幾何」の5.1章「関数を作ろう」を参考にすること。関数を自分で作ると、より複雑な計算もできるようになる。

さて、A_2特異点の原点での爆発を計算してみよう。

```
i5 : R = QQ[x,y,z]/(x*y-z^3)

o5 = R

o5 : QuotientRing

i6 : A = reesAlgebra ideal(x,y,z)

o6 = A

o6 : QuotientRing
```

スキーム論を知っている人向けの説明：環RのイデアルIに対し、次数付き環Rees環が以下で定義される。

$$A := R[It] = \bigoplus_{n \geq 0} I^n$$

これのProjが$\mathrm{Spec}\, R$のイデアルIによる爆発となる。Iの生成元$f_1, \dots, f_l$を固定すると、$R[x_1, \dots, x_l]$の商環$R[x_1, \dots, x_l]/J$として書ける。爆発はl個のアフィン開集合で覆われ、各$i \in \{1, \dots, l\}$に対し、

$$B_i := R[x_1, \dots, x_l]/(J + (x_i - 1))$$

がアフィン開集合の座標環となる。

A_2特異点の原点での爆発が非特異かどうか確認する。

```
7 : isBlowupSmooth A

o7 = true
```

この計算でA_2特異点の原点での爆発が非特異で、特異点解消を与えていることが分かった。

各アフィン開集合を計算すると：

```
i8 : affineCharts A

                                    QQ[w ..w , x..z]
                                        0   2
o8 = {-----------------------------------------------------
          3                                            2
      (- z  + x*y, w x - w y, w y - w z, w x - w z, w z  - w 
                    1     2    0     1    0     2    0      2
     -----------------------------------------------------
                                   QQ[w ..w , x..z]
```

```
                                   0   2
      --------------------------------------------------------
         3                                             2
      (- z  + x*y, w x - w y, w y - w z, w x - w z, w z  - w y
                    1     2    0     1    0     2    0      2
      --------------------------------------------------------
                               QQ[w ..w , x..z]
                                   0   2
      --------------------------------------------------------
         3                                             2
      (- z  + x*y, w x - w y, w y - w z, w x - w z, w z  - w y
                    1     2    0     1    0     2    0      2

o8 : List

i9 : singularLocus o8_0 -- 一つ目の座標環のsingular locusを計算



o9 = --------------------------------------------------------
         3                                             2
      (- z  + x*y, w x - w y, w y - w z, w x - w z, w z  - w y
                    1     2    0     1    0     2    0      2

o9 : QuotientRing

i10 : minimalPresentation o9 -- minimalPresentationで環の表示を

       QQ[]
o10 = ----   -- singular locusは空集合
        1

o10 : QuotientRing
```

爆発の例外集合は以下のように計算できる。

```
i11 : specialFiber ideal(x,y,z)

      QQ[w ..w ]
          0   2
o11 = ----------
         w w
          0 1

o11 : QuotientRing
```

例外集合は斉次座標w_0, w_1, w_2を持つ射影平面$\mathbb{P}^2$の中で

$$w_0 w_1 = 1$$

で定義されることが分かった。これより、例外集合は2つの射影直線が1点で直交していることが分かる。

演習

1. A_3特異点、D_4特異点で同様の計算をしてみよう。
2. 時間が余った人は、「Macaulay2による計算代数幾何」を見て、いろいろな計算を試してみよう。今後もM2を使う可能性のある人は、5.1章「関数を作ろう」を読んで、関数の作り方を学習しよう。